
**************
*** Fig. 2 ***
**************

* load frame
frame change comparative_data

* dummy economic growth	
reg main_out ///
	i.income_loss##i.declining##c.region_talk_pol ///
	i.class2 i.class3 ///
	i.not_working ///
	age ///
	i.edu2 i.edu3 i.edu4 i.edu5 ///
	i.gender ///
	i.cntryn ///
	, ///
		cluster(region_iso2)
		
margins declining, dydx(income_loss) at(region_talk_pol=(30(10)70))
mplotoffset,  offset(0.5) recast(scatter) ///
	title("") ///
	ytitle("Mainstream vote", size(large)) ///
	xtitle("People who talk politics", size(large)) ///		
	yline(0, lw(thin) lstyle(foreground) ) ///
	ylabel(, grid glcolor(black%70) glw(thin) labs(medl)) ///
	xlabel(, nogrid) ///
	aspect(1) ///
	plot1opts(msymbol(D) msize(medium) mcolor(red) lcolor(red) lwidth(medthick)) ci1opts(lcolor(red)) ///
	plot2opts(msymbol(O) msize(medium) mcolor(blue) lp(solid)  mlcolor(blue) lcolor(blue) lwidth(medthick)) ci2opts(lcolor(blue)) ///
	legend( ///
		region(lwidth(none)) ///
		col(2) ///
		position(6) ///
		order(3 "Growing region" 4 "Declining region"))
		
graph addplot histogram region_talk_pol, ///
	yaxis(2) ///
	yscale(range(0 0.5) off axis(2)) ///
	xlabel(10(10)80) ///
	color(black%10) ///
	legend( ///
		order(3 "Growing region" 4 "Declining region"))
graph export "$folder/replication/figures/fig2a.pdf", replace
	
* three categories of economic growth	
reg main_out ///
	i.income_loss##i.growth_three##c.region_talk_pol ///
	i.class2 i.class3 ///
	i.not_working ///
	age ///
	i.edu2 i.edu3 i.edu4 i.edu5 ///
	i.gender ///
	i.cntryn ///
	, ///
		cluster(region_iso2)
		
margins growth_three, dydx(income_loss) at(region_talk_pol=(30(10)70))
mplotoffset,  offset(0.5) recast(scatter) ///
		title("") ///
		ytitle("Mainstream vote", size(large)) ///
		xtitle("People who talk politics", size(large)) ///		
		yline(0, lw(thin) lstyle(foreground) ) ///
		ylabel(, grid glcolor(black%70) glw(thin) labs(medl)) ///
		xlabel(, nogrid) ///
		aspect(1) ///
		plot1opts(msymbol(D) msize(medium) mcolor(red) lcolor(red) lwidth(medthick)) ci1opts(lcolor(red)) ///
		plot2opts(msymbol(O) msize(medium) mcolor(orange) lp(solid)  mlcolor(orange) lcolor(orange) lwidth(medthick)) ci2opts(lcolor(orange)) ///
		plot3opts(msymbol(O) msize(medium) mcolor(blue) lp(solid)  mlcolor(blue) lcolor(blue) lwidth(medthick)) ci3opts(lcolor(blue)) ///
		legend( ///
			region(lwidth(none)) ///
			col(3) ///
			position(6) ///
			order(4 "GDP growth < 0" 5 "GDP growth 0-2" 6 "GDP growth > 2"))
graph addplot histogram region_talk_pol, ///
	yaxis(2) ///
	yscale(range(0 0.5) off axis(2)) ///
	xlabel(10(10)80) ///
	color(black%10) ///
	legend( ///
		order(4 "GDP growth < 0" 5 "GDP growth 0-2" 6 "GDP growth > 2"))

graph export "$folder/replication/figures/fig2b.pdf", replace
	

